UNIVERSIDADE FEDERAL DO RIO GRANDE DO SUL
INSTITUTO DE MATEMÁTICA
DEPARTAMENTO DE MATEMÁTICA PURA E APLICADA
Disciplina: MAT01168 -Matemática Aplicada -Semestre Letivo 2008/2
Professoras: Elisabeta Gallicchio e Irene Strauch
TERCEIRA ÁREA
SÉRIES DE FOURIER UTILIZANDO MAPLE

Série de Fourier Complexa ou Série de Fourier Exponencial

Cálculo da série de Fourier Complexa de uma função f(t) periódica, ou seja, f(t+T) = f(t), **com período T=2L, integrável no intervalo simétrico [-L,L].**

Sendo $w_0 = \frac{\pi}{L}$ ou $w_n = \frac{2n\pi}{T}$, a função f(t) é representada pela série $f(t) = \sum_{n=-\infty}^{\infty} C_n e^{\frac{jn\pi t}{L}}$ ou $f(t) = \sum_{n=-\infty}^{\infty} C_n e^{jnw_0 t}$ ou, ainda, $f(t) = \sum_{n=-\infty}^{\infty} C_n e^{jw_n t}$

onde os coeficientes Cn são dados por:

$$C_n = \frac{1}{2L}\left(\int_{-L}^{L} f(t)\, e^{-jnw_0 t}\, dt\right) \quad \text{ou, para n=0,} \quad \text{Co=} \frac{a_0}{2} = \frac{1}{2L}\left(\int_{-L}^{L} f(t)\, dt\right)$$

Deve-se observar que $C_n = \frac{1\,(a_n - j\,b_n)}{2}$ e $C_{-n} = \frac{1\,(a_n + j\,b_n)}{2}$, isto é , C_{-n} é o conjugado de C_n, e os coeficiente da Série de Fourier Trigonométrica

$a_n = 2\,\mathbf{R}\{C_n\}$ e $b_n = -2\,Im\,\{C_n\}$.

Aqui, a unidade complexa foi denotada por j $=\sqrt{-1}$ e, a seguir, nos comandos do Maple, deverá ser escrita como I$=\sqrt{-1}$.

Ortogonalidade

Deve-se mostrar que o conjunto de funções {$e^{-jnw_0 t}$, $e^{jnw_0 t}$}, n=1,2,3.. forma uma base ortogonal, no intervalo [-L,L], com o produto interno $\langle \Phi_m, \Phi_n \rangle = \int_{-L}^{L} \Phi_m(t)\, \Phi_n(t)\, dt$ =0, m<>n . E, para a norma quadrática, n=m, que $\|\Phi_n(x)\|$ ^2 $= \int_{-L}^{L} e^{\frac{In\pi t}{L}}\, e^{-\frac{In\pi t}{L}}\, dt$=2L

Deve ser declarado m e n são números inteiros e distintos.

```
> restart:assume(m,integer);assume(n,integer);interface
  (showassumed=0):(m<>n);interface(showassumed=0):#Para a
  simplificação dos coeficientes, explicita-se que m,n são números
```

```
inteiros e diferentes
```

$$m\sim \neq n\sim \quad (1.1)$$

```
> prod_interno:=simplify(int(exp(I*n*Pi*t/L)*exp(-I*m*Pi*t/L),t=-L..L));
> Norma_quad:=int(exp(I*n*Pi*t/L)*exp(-I*n*Pi*t/L),t=-L..L);
```

$$prod\_interno := 0 \quad (1.2)$$

$$Norma\_quad := 2\,L$$

Exemplo 1

Determinar a série de Fourier Complexa da função f(t)=exp(t), -π<t<π sendo f(t+2π)=f(t)

Solução:

```
> restart:with(plots):with(plottools):with(student):assume(n,integer):interface(showassumed=0):
Warning, the name changecoords has been redefined

Warning, the name arrow has been redefined
```

```
> T:=2*Pi;L:=T/2;w[0]:=2*Pi/T;w[n]:=n*w[0];
```

$$T := 2\,\pi \quad (2.1)$$

$$L := \pi$$

$$w_0 := 1$$

$$w_{n\sim} := n\sim$$

Definir a função periódica a ser expandida em série de Fourier

```
> f(t):=exp(t);
```

$$f(t) := e^{t} \quad (2.2)$$

Traçar o gráfico da função f(t) definida acima

```
> plot(f(t),t=-L..L,thickness=3,scaling=constrained,discont=true,titlefont=[COURIER,DEFAULT,16],labels=["t","f(t)"],labelfont=[COURIER,DEFAULT,16],axesfont=[COURIER,DEFAULT,16],title=`Gráfico de f(t)`);
```

Gr�fico de f(t)



Determinar os coeficientes $C_n = < f(t), \Phi_n > /|| \Phi_n||$^2 da série de Fourier

```
> C[n]:=simplify((1/(2*L)*int(f(t)*exp(-I*w[n]*t),t=-L..L)));
```

$$C_{n\sim} := \frac{1}{2} \frac{(-1)^{1+n\sim} (e^{2\pi} - 1) e^{-\pi}}{\pi(-1 + n\sim I)} \quad (2.3)$$

```
> Cn:=simplify(combine(convert(C[n],trig)),trig);
```

$$Cn := -\frac{(-1)^{n\sim} \sinh(\pi)}{(-1 + n\sim I)\pi} \quad (2.4)$$

Racionalizando e simplificando o denominador com o comando "radsimp"

```
> C[n]:=radsimp(evalc(Cn)*(-1-n*I)/((-1-n*I)));
```

$$C_{n\sim} := \frac{(-1)^{n\sim} \sinh(\pi) (1 + n\sim I)}{(1 + n\sim^2)\pi} \quad (2.5)$$

Calcular o coeficiente C_0

```
> C[0]:= simplify(1/(2*L)*int(f(t),t=-L..L));C[0]:=combine(convert
  (C[0],trig));
```

$$C_0 := \frac{1}{2} \frac{(e^{2\pi} - 1) e^{-\pi}}{\pi} \quad (2.6)$$

$$C_0 := \frac{\sinh(\pi)}{\pi}$$

```
> serie_complexa_f(t)=Sum(C[n]*exp(I*n*Pi/L*t),n=-infinity..
  infinity);
```

$$serie\_complexa\_f(t) = \sum_{n\sim = -\infty}^{\infty} \frac{(-1)^{n\sim} \sinh(\pi) (1 + n\sim I) e^{n\sim t I}}{(1 + n\sim^2)\pi} \quad (2.7)$$

ou, explicitando C_0

```
> serie_complexa_f(t)= C[0]+Sum(C[n]*exp(I*n*Pi/L*t),n=-infinity..
  -1)+Sum(C[n]*exp(I*n*Pi/L*t),n=1..infinity);
```

(2.8)

$$serie\_complexa\_f(t) = \frac{\sinh(\pi)}{\pi} + \sum_{n\sim=-\infty}^{-1} \frac{(-1)^{n\sim} \sinh(\pi)\ (1 + n\sim \mathrm{I})\ e^{n\sim t \mathrm{I}}}{(1 + n\sim^2)\ \pi} + \sum_{n\sim=1}^{\infty} \frac{(-1)^{n\sim} \sinh(\pi)\ (1 + n\sim \mathrm{I})\ e^{n\sim t \mathrm{I}}}{(1 + n\sim^2)\ \pi} \qquad \textbf{(2.8)}$$

GRÁFICO

Para o gráfico, deve-se obter a Série Fourier Trigonométrica . Deste modo, vamos calcular a série real para n=-10..10, porque os termos em -n também devem ser adicionados. Então, trocando Sum por sum para expandir o somatório:

```
> fap10(t):= sum(C[n]*exp(I*n*Pi/L*t),n=-10..10);
```

$$fap10(t) := \frac{\left(-\frac{1}{82} + \frac{9}{82}\mathrm{I}\right)\sinh(\pi)\ e^{-9\mathrm{I}t}}{\pi} + \frac{\left(-\frac{1}{2} - \frac{1}{2}\mathrm{I}\right)\sinh(\pi)\ e^{t\mathrm{I}}}{\pi} \qquad \textbf{(2.9)}$$
$$+ \frac{\left(\frac{1}{65} - \frac{8}{65}\mathrm{I}\right)\sinh(\pi)\ e^{-8\mathrm{I}t}}{\pi} + \frac{\left(\frac{1}{5} + \frac{2}{5}\mathrm{I}\right)\sinh(\pi)\ e^{2\mathrm{I}t}}{\pi}$$
$$+ \frac{\left(-\frac{1}{50} + \frac{7}{50}\mathrm{I}\right)\sinh(\pi)\ e^{-7\mathrm{I}t}}{\pi} + \frac{\left(-\frac{1}{10} - \frac{3}{10}\mathrm{I}\right)\sinh(\pi)\ e^{3\mathrm{I}t}}{\pi}$$
$$+ \frac{\left(\frac{1}{37} - \frac{6}{37}\mathrm{I}\right)\sinh(\pi)\ e^{-6\mathrm{I}t}}{\pi} + \frac{\left(-\frac{1}{26} + \frac{5}{26}\mathrm{I}\right)\sinh(\pi)\ e^{-5\mathrm{I}t}}{\pi}$$
$$+ \frac{\left(\frac{1}{17} + \frac{4}{17}\mathrm{I}\right)\sinh(\pi)\ e^{4\mathrm{I}t}}{\pi} + \frac{\left(-\frac{1}{82} - \frac{9}{82}\mathrm{I}\right)\sinh(\pi)\ e^{9\mathrm{I}t}}{\pi} + \frac{\sinh(\pi)}{\pi}$$
$$+ \frac{\left(\frac{1}{101} + \frac{10}{101}\mathrm{I}\right)\sinh(\pi)\ e^{10\mathrm{I}t}}{\pi} + \frac{\left(\frac{1}{101} - \frac{10}{101}\mathrm{I}\right)\sinh(\pi)\ e^{-10\mathrm{I}t}}{\pi}$$
$$+ \frac{\left(\frac{1}{17} - \frac{4}{17}\mathrm{I}\right)\sinh(\pi)\ e^{-4\mathrm{I}t}}{\pi} + \frac{\left(-\frac{1}{50} - \frac{7}{50}\mathrm{I}\right)\sinh(\pi)\ e^{7\mathrm{I}t}}{\pi}$$
$$+ \frac{\left(-\frac{1}{10} + \frac{3}{10}\mathrm{I}\right)\sinh(\pi)\ e^{-3\mathrm{I}t}}{\pi} + \frac{\left(-\frac{1}{26} - \frac{5}{26}\mathrm{I}\right)\sinh(\pi)\ e^{5\mathrm{I}t}}{\pi}$$
$$+ \frac{\left(\frac{1}{37} + \frac{6}{37}\mathrm{I}\right)\sinh(\pi)\ e^{6\mathrm{I}t}}{\pi} + \frac{\left(\frac{1}{5} - \frac{2}{5}\mathrm{I}\right)\sinh(\pi)\ e^{-2\mathrm{I}t}}{\pi}$$

$$+\frac{\left(-\frac{1}{2}+\frac{1}{2}\,\mathrm{I}\right)\sinh(\pi)\,\mathrm{e}^{-\mathrm{I}\,t}}{\pi}+\frac{\left(\frac{1}{65}+\frac{8}{65}\,\mathrm{I}\right)\sinh(\pi)\,\mathrm{e}^{8\,\mathrm{I}\,t}}{\pi}$$

Agora vamos converter a série acima para a forma real (observe os termos em seno e co-seno)

```
> serie_real:=evalc(Re(fap10(t)));
```

$$serie\_real := \frac{\sinh(\pi)}{\pi}-\frac{1}{41}\,\frac{\sinh(\pi)\cos(9\,t)}{\pi}+\frac{9}{41}\,\frac{\sinh(\pi)\sin(9\,t)}{\pi}-\frac{\sinh(\pi)\cos(t)}{\pi}+\frac{\sinh(\pi)\sin(t)}{\pi}+\frac{2}{65}\,\frac{\sinh(\pi)\cos(8\,t)}{\pi}-\frac{16}{65}\,\frac{\sinh(\pi)\sin(8\,t)}{\pi}+\frac{2}{5}\,\frac{\sinh(\pi)\cos(2\,t)}{\pi}-\frac{4}{5}\,\frac{\sinh(\pi)\sin(2\,t)}{\pi}-\frac{1}{25}\,\frac{\sinh(\pi)\cos(7\,t)}{\pi}+\frac{7}{25}\,\frac{\sinh(\pi)\sin(7\,t)}{\pi}-\frac{1}{5}\,\frac{\sinh(\pi)\cos(3\,t)}{\pi}+\frac{3}{5}\,\frac{\sinh(\pi)\sin(3\,t)}{\pi}+\frac{2}{37}\,\frac{\sinh(\pi)\cos(6\,t)}{\pi}-\frac{12}{37}\,\frac{\sinh(\pi)\sin(6\,t)}{\pi}-\frac{1}{13}\,\frac{\sinh(\pi)\cos(5\,t)}{\pi}+\frac{5}{13}\,\frac{\sinh(\pi)\sin(5\,t)}{\pi}+\frac{2}{17}\,\frac{\sinh(\pi)\cos(4\,t)}{\pi}-\frac{8}{17}\,\frac{\sinh(\pi)\sin(4\,t)}{\pi}+\frac{2}{101}\,\frac{\sinh(\pi)\cos(10\,t)}{\pi}-\frac{20}{101}\,\frac{\sinh(\pi)\sin(10\,t)}{\pi} \quad (2.10)$$

Traçar os gráficos da função dada e da sua aproximação em série de Fourier truncada.

```
> plot({f(t),serie_real},t=-L..L,scaling=constrained,titlefont=
  [COURIER,DEFAULT,16],labels=["t","f(t)"],labelfont=[COURIER,
  DEFAULT,16],axesfont=[COURIER,DEFAULT,16],title=`f(t) e a Série
  de Fourier para n=-10..10`, thickness=2, color=[aquamarine,blue])
  ;
```



Somente para conferir os cálculos efetuados acima:

Vamos ,agora, calcular os coeficientes da **série real** com as fórmulas: $a_n := 2\,\Re(C_n)$,

$b_n = -2\,\Im(C_n)$ e $\frac{a_0}{2} = C_0$ e o comando "convert" para a forma real e imaginária.

```
> a[n]:=simplify(convert(evalc(2*Re(Cn)),exp)):b[n]:=simplify
  (convert(evalc(-2*Im(Cn)),exp)):a[0]:=simplify(convert(2*subs(n=
  0,Cn),exp)): print(an=a[n],bn=b[n],a0=a[0]);
```

$$an = \frac{\left(e^{2\pi}-1\right)(-1)^{n\sim}\,e^{-\pi}}{\pi\left(1+n\sim^2\right)},\ bn = -\frac{(-1)^{n\sim}\left(e^{2\pi}-1\right)n\sim\,e^{-\pi}}{\pi\left(1+n\sim^2\right)},\ a0 = \frac{\left(e^{2\pi}-1\right)e^{-\pi}}{\pi} \qquad \textbf{(2.11)}$$

FINALMENTE: **para conferir os valores destes coeficientes e a série real, acima obtida, usaremos as fórmulas de Euler-Fourier:**

$$a_n = \frac{1}{L}\left(\int_{-L}^{L} f(t)\cos\left(\frac{n\,\pi\,t}{L}\right)dt\right) \qquad \frac{a_0}{2} = \frac{1}{2\,L}\left(\int_{-L}^{L} f(t)\,dt\right)$$

$$b_n = \frac{1}{L}\left(\int_{-L}^{L} f(t)\sin\left(\frac{n\,\pi\,t}{L}\right)dt\right)$$

Cálculo dos coeficientes a_n, n=1,2,3...

```
> an:=simplify(int(f(t)*cos(n*Pi/L*t),t=-L..L)/L);
```

$$an := \frac{(-1)^{n\sim}\left(e^{2\pi}-1\right)e^{-\pi}}{\left(1+n\sim^2\right)\pi} \qquad \textbf{(2.12)}$$

O coeficiente a_0

```
> a0:= simplify(int(f(t),t=-L..L)/L);
```

$$a0 := \frac{\left(e^{2\pi}-1\right)e^{-\pi}}{\pi} \qquad \textbf{(2.13)}$$

A seguir, os coeficientes b_n, para n=1,2,3,...

```
> bn:=simplify(int(f(t)*sin(n*Pi/L*t),t=-L..L)/L);
```

$$bn := -\frac{(-1)^{n\sim}\left(e^{2\pi}-1\right)n\sim\,e^{-\pi}}{\left(1+n\sim^2\right)\pi} \qquad \textbf{(2.14)}$$

CONFIRA ESTES RESULTADOS com os obtidos acima! A série trigonométrica é dada por

```
> serie_trig:=a0/2+ sum(an*cos(n*Pi/L*t)+bn*sin(n*Pi/L*t),n=1..
  infinity);
```

$$serie\_trig := \frac{1}{2}\,\frac{\left(e^{2\pi}-1\right)e^{-\pi}}{\pi} + \sum_{n\sim=1}^{\infty}\left(\frac{(-1)^{n\sim}\left(e^{2\pi}-1\right)e^{-\pi}\cos(n\sim t)}{\left(1+n\sim^2\right)\pi} - \frac{(-1)^{n\sim}\left(e^{2\pi}-1\right)n\sim\,e^{-\pi}\sin(n\sim t)}{\left(1+n\sim^2\right)\pi}\right) \qquad \textbf{(2.15)}$$

Aproximar a função f(t) com a série truncada n=0..10 termos.

```
> f_ap(t):=a0/2+ sum(an*cos(w[n]*t)+bn*sin(w[n]*t),n=1..10);
```

$$f\_ap(t) := \frac{1}{2}\,\frac{\left(e^{2\pi}-1\right)e^{-\pi}}{\pi} - \frac{1}{2}\,\frac{\left(e^{2\pi}-1\right)e^{-\pi}\cos(t)}{\pi} + \frac{1}{2}\,\frac{\left(e^{2\pi}-1\right)e^{-\pi}\sin(t)}{\pi}$$
$$+ \frac{1}{5}\,\frac{\left(e^{2\pi}-1\right)e^{-\pi}\cos(2\,t)}{\pi} - \frac{2}{5}\,\frac{\left(e^{2\pi}-1\right)e^{-\pi}\sin(2\,t)}{\pi}$$
$$- \frac{1}{10}\,\frac{\left(e^{2\pi}-1\right)e^{-\pi}\cos(3\,t)}{\pi} + \frac{3}{10}\,\frac{\left(e^{2\pi}-1\right)e^{-\pi}\sin(3\,t)}{\pi}$$
$$+ \frac{1}{17}\,\frac{\left(e^{2\pi}-1\right)e^{-\pi}\cos(4\,t)}{\pi} - \frac{4}{17}\,\frac{\left(e^{2\pi}-1\right)e^{-\pi}\sin(4\,t)}{\pi}$$
$$- \frac{1}{26}\,\frac{\left(e^{2\pi}-1\right)e^{-\pi}\cos(5\,t)}{\pi} + \frac{5}{26}\,\frac{\left(e^{2\pi}-1\right)e^{-\pi}\sin(5\,t)}{\pi}$$
$$+ \frac{1}{37}\,\frac{\left(e^{2\pi}-1\right)e^{-\pi}\cos(6\,t)}{\pi} - \frac{6}{37}\,\frac{\left(e^{2\pi}-1\right)e^{-\pi}\sin(6\,t)}{\pi}$$
$$- \frac{1}{50}\,\frac{\left(e^{2\pi}-1\right)e^{-\pi}\cos(7\,t)}{\pi} + \frac{7}{50}\,\frac{\left(e^{2\pi}-1\right)e^{-\pi}\sin(7\,t)}{\pi}$$
$$+ \frac{1}{65}\,\frac{\left(e^{2\pi}-1\right)e^{-\pi}\cos(8\,t)}{\pi} - \frac{8}{65}\,\frac{\left(e^{2\pi}-1\right)e^{-\pi}\sin(8\,t)}{\pi}$$
$$- \frac{1}{82}\,\frac{\left(e^{2\pi}-1\right)e^{-\pi}\cos(9\,t)}{\pi} + \frac{9}{82}\,\frac{\left(e^{2\pi}-1\right)e^{-\pi}\sin(9\,t)}{\pi}$$
$$+ \frac{1}{101}\,\frac{\left(e^{2\pi}-1\right)e^{-\pi}\cos(10\,t)}{\pi} - \frac{10}{101}\,\frac{\left(e^{2\pi}-1\right)e^{-\pi}\sin(10\,t)}{\pi} \tag{2.16}$$

Vamos traçar os gráficos da função dada e da sua aproximação em série de Fourier truncada.

```
> plot({f(t),f_ap(t)},t=-L..L,scaling=constrained,titlefont=
  [COURIER,DEFAULT,16],labels=["t","f(t)"],labelfont=[COURIER,
  DEFAULT,16],axesfont=[COURIER,DEFAULT,16],title=`f(t) e a Série
  de Fourier Trigonométrica, para n=0..10`, thickness=2, color=
  [aquamarine,blue]);
```

f(t) e a S�rie de Fourier Trigonom�trica, para n=0..10



==
=========

Espectro de freqüência (espectro de amplitude) - exemplo1 :

é o conjunto de pontos $(n w_0, |C_n|)$ ou $(w_n, |C_n|)$, onde n é um número inteiro.

Cálculo das amplitudes, para o espectro de freqüência

```
> C[n];moduloCn:=abs(C[n]);
```

$$\frac{(-1)^{n\sim} \sinh(\pi)\,(1 + n\sim \mathrm{I})}{(1 + n\sim^2)\,\pi} \quad \textbf{(3.1)}$$

$$moduloCn := \frac{\sinh(\pi)\,|(-1)^{n\sim}|}{\sqrt{1 + n\sim^2}\,\pi}$$

```
> for i from -10 to 10 do C[i]:=evalf(subs(n=i,moduloCn)) :  w[i]:=
  subs(n=i,w[n]):od:
> for k from -10 to 10 do lprint(frequencia[k]=w[k],amplitude[k]=C
  [k]):od;
```

```
frequencia[-10] = -10, amplitude[-10] = .3657834234
frequencia[-9] = -9, amplitude[-9] = .4059548937
frequencia[-8] = -8, amplitude[-8] = .4559613479
frequencia[-7] = -7, amplitude[-7] = .5198759238
frequencia[-6] = -6, amplitude[-6] = .6043434857
frequencia[-5] = -5, amplitude[-5] = .7209381923
frequencia[-4] = -4, amplitude[-4] = .8915798541
frequencia[-3] = -3, amplitude[-3] = 1.162477905
frequencia[-2] = -2, amplitude[-2] = 1.643992020
frequencia[-1] = -1, amplitude[-1] = 2.599379618
frequencia[0] = 0, amplitude[0] = 3.676077911
frequencia[1] = 1, amplitude[1] = 2.599379618
frequencia[2] = 2, amplitude[2] = 1.643992020
frequencia[3] = 3, amplitude[3] = 1.162477905
frequencia[4] = 4, amplitude[4] = .8915798541
frequencia[5] = 5, amplitude[5] = .7209381923
```

```
frequencia[6] = 6, amplitude[6] = .6043434857
frequencia[7] = 7, amplitude[7] = .5198759238
frequencia[8] = 8, amplitude[8] = .4559613479
frequencia[9] = 9, amplitude[9] = .4059548937
frequencia[10] = 10, amplitude[10] = .3657834234
```

```
> for i from -10 to 10 do g[i]:=line([w[i],0],[w[i],C[i]], color=
  red, thickness=3):od:
```

```
> Graf2:=plots[display]({g[0],g[-1],g[-2],g[-3],g[-4],g[-5],g[-6],g
  [-7],g[-8],g[-9],g[-10],g[1],g[2],g[3],g[4],g[5],g[6],g[7],g[8],g
  [9],g[10]},xtickmarks=[-10,-9,-8,-7,-6,-5,-4,-3,-2,-1,0,1,2,3,4,
  5,6,7,8,9,10],titlefont=[COURIER,DEFAULT,12],labels=["w[n]","|C
  [n]|"],labelfont=[COURIER,DEFAULT,10],axesfont=[COURIER,DEFAULT,
  10],title=`Espectro de Amplitude,n=-10..10`):Graf2;
```



==
==========================

Exemplo 2

Determinar a série de Fourier Complexa da função f(t)= $\frac{3\,t}{4}$, 0<t<8, sendo f(t)=f(t+8)

Solução:

```
> restart:with(plots):with(plottools):with(student):assume(n,
  integer):interface(showassumed=0):
Warning, the name changecoords has been redefined

Warning, the name arrow has been redefined
```

```
> T:=8;L:=T/2;w[0]:=2*Pi/T;w[n]:=n*w[0];
```

$$T := 8 \tag{4.1}$$

$$L := 4$$

$$w_0 := \frac{\pi}{4}$$

$$w_{n\sim} := \frac{n\sim \pi}{4}$$

Definir a função periódica a ser expandida em série de Fourier

```
> f(t):=(3/4)*t;
```

$$f(t) := \frac{3\,t}{4} \tag{4.2}$$

Traçar o gráfico da função f(t) definida acima

```
> plot(f(t),t=0..T,thickness=3,scaling=constrained,discont=true,
  titlefont=[COURIER,DEFAULT,16],labels=["t","f(t)"],labelfont=
  [COURIER,DEFAULT,16],axesfont=[COURIER,DEFAULT,16],title=`f(t)=
  3t/4`);
```



Determinar os coeficientes $C_n = < f(t), \Phi_n > / || \Phi_n ||$^2 da série de Fourier (agora com o cálculo das integrais no período T).

```
> C[n]:=simplify((1/(2*L)*int(f(t)*exp(-I*w[n]*t),t=0..T)));
```

$$C_{n\sim} := \frac{3\,I}{n\sim \pi} \tag{4.3}$$

```
> serie_complexa_f(t)=Sum(C[n]*exp(I*w[n]*t),n=-infinity..infinity)
  ;
```

$$serie\_complexa\_f(t) = \sum_{n\sim = -\infty}^{\infty} \frac{3\,I\,e^{\frac{1}{4}\,I\,n\sim \pi\,t}}{n\sim \pi} \tag{4.4}$$

Para o gráfico, deve-se obter a Série Fourier Trigonométrica . Deste modo, vamos calcular a série real para n=-10..10, porque os termos em -n também devem ser adicionados. Então, trocando Sum por sum para expandir o somatório: ATENÇÃO para a resposta do Maple

```
> fap10(t):= sum(C[n]*exp(I*w[n]*t),n=-10..10);
Error, (in sum) numeric exception: division by zero
```

O termo com n=0 , C_0 apresenta divisão por zero. Vamos calcular o coeficienteC[0]

```
> C[0]:= simplify(1/(2*L)*int(f(t),t=0..T));
```

$$C_0 := 3 \tag{4.5}$$

```
> serie_complexa_f(t)=Sum(C[n]*exp(I*w[n]*t),n=-infinity..-1)+C[0]+
  Sum(C[n]*exp(I*w[n]*t),n=1..infinity);
```

$$serie\_complexa\_f(t) = \sum_{n\sim=-\infty}^{-1} \frac{3\,\mathrm{I}\,\mathrm{e}^{\frac{1}{4}\mathrm{I}n\sim\pi t}}{n\sim\pi} + 3 + \sum_{n\sim=1}^{\infty} \frac{3\,\mathrm{I}\,\mathrm{e}^{\frac{1}{4}\mathrm{I}n\sim\pi t}}{n\sim\pi} \tag{4.6}$$

```
> serie_comp10:=sum(C[n]*exp(I*w[n]*t),n=-10..-1)+3+sum(C[n]*exp(I*
  w[n]*t),n=1..10);
```

$$\begin{aligned} serie\_comp10 :=& -\frac{\frac{1}{2}\,\mathrm{I}\,\mathrm{e}^{-\frac{3}{2}\mathrm{I}\pi t}}{\pi} - \frac{3\,\mathrm{I}\,\mathrm{e}^{-\frac{1}{4}\mathrm{I}\pi t}}{\pi} + \frac{\frac{3}{7}\,\mathrm{I}\,\mathrm{e}^{\frac{7}{4}\mathrm{I}\pi t}}{\pi} + \frac{\frac{3}{10}\,\mathrm{I}\,\mathrm{e}^{\frac{5}{2}\mathrm{I}\pi t}}{\pi} \\ & - \frac{\frac{3}{2}\,\mathrm{I}\,\mathrm{e}^{-\frac{1}{2}\mathrm{I}\pi t}}{\pi} + \frac{3\,\mathrm{I}\,\mathrm{e}^{\frac{1}{4}\mathrm{I}\pi t}}{\pi} - \frac{\frac{3}{7}\,\mathrm{I}\,\mathrm{e}^{-\frac{7}{4}\mathrm{I}\pi t}}{\pi} + \frac{\frac{3}{5}\,\mathrm{I}\,\mathrm{e}^{\frac{5}{4}\mathrm{I}\pi t}}{\pi} + \frac{\mathrm{e}^{\frac{3}{4}\mathrm{I}\pi t}\,\mathrm{I}}{\pi} + \frac{\frac{3}{8}\,\mathrm{I}\,\mathrm{e}^{2\mathrm{I}\pi t}}{\pi} \\ & + 3 - \frac{\frac{3}{8}\,\mathrm{I}\,\mathrm{e}^{-2\mathrm{I}\pi t}}{\pi} - \frac{\mathrm{I}\,\mathrm{e}^{-\frac{3}{4}\mathrm{I}\pi t}}{\pi} + \frac{\frac{3}{4}\,\mathrm{I}\,\mathrm{e}^{\pi t\mathrm{I}}}{\pi} - \frac{\frac{3}{4}\,\mathrm{I}\,\mathrm{e}^{-\mathrm{I}\pi t}}{\pi} - \frac{\frac{1}{3}\,\mathrm{I}\,\mathrm{e}^{-\frac{9}{4}\mathrm{I}\pi t}}{\pi} + \frac{\frac{3}{2}\,\mathrm{I}\,\mathrm{e}^{\frac{1}{2}\mathrm{I}\pi t}}{\pi} \\ & - \frac{\frac{3}{10}\,\mathrm{I}\,\mathrm{e}^{-\frac{5}{2}\mathrm{I}\pi t}}{\pi} - \frac{\frac{3}{5}\,\mathrm{I}\,\mathrm{e}^{-\frac{5}{4}\mathrm{I}\pi t}}{\pi} + \frac{\frac{1}{2}\,\mathrm{I}\,\mathrm{e}^{\frac{3}{2}\mathrm{I}\pi t}}{\pi} + \frac{\frac{1}{3}\,\mathrm{I}\,\mathrm{e}^{\frac{9}{4}\mathrm{I}\pi t}}{\pi} \end{aligned} \tag{4.7}$$

Agora vamos converter a série acima para a forma real (observe os termos em seno e co-seno)

```
> serie_real:=expand(evalc(Re(serie_comp10)));
```

$$\begin{aligned} serie\_real :=& 3 - \frac{\sin\left(\frac{3\pi t}{2}\right)}{\pi} - \frac{6\sin\left(\frac{\pi t}{4}\right)}{\pi} - \frac{6}{7}\,\frac{\sin\left(\frac{7\pi t}{4}\right)}{\pi} - \frac{3}{5}\,\frac{\sin\left(\frac{5\pi t}{2}\right)}{\pi} \\ & - \frac{3\sin\left(\frac{\pi t}{2}\right)}{\pi} - \frac{6}{5}\,\frac{\sin\left(\frac{5\pi t}{4}\right)}{\pi} - \frac{2\sin\left(\frac{3\pi t}{4}\right)}{\pi} - \frac{3}{2}\,\frac{\sin(\pi t)\cos(\pi t)}{\pi} \\ & - \frac{3}{2}\,\frac{\sin(\pi t)}{\pi} - \frac{2}{3}\,\frac{\sin\left(\frac{9\pi t}{4}\right)}{\pi} \end{aligned} \tag{4.8}$$

Traçar os gráficos da função dada e da sua aproximação em série de Fourier truncada.

```
> plot({f(t),serie_real},t=0..T,scaling=constrained,titlefont=
  [COURIER,DEFAULT,16],labels=["t","f(t)"],labelfont=[COURIER,
  DEFAULT,16],axesfont=[COURIER,DEFAULT,16],title=`f(t) e a Série
  de Fourier para n=-10..10`, thickness=2, color=[aquamarine,blue])
  ;
```



Vamos ,agora, calcular os coeficientes da **série real** com as fórmulas: $a_n := 2\,\Re(C_n)$,

$b_n = -2\,\Im(C_n)$ e $\frac{a_0}{2} = C_0$ e o comando "convert" para a forma real e imaginária.

```
> a[n]:=evalc(2*Re(C[n])):b[n]:=evalc(-2*Im(C[n])):a[0]:=2*C[0]:
  print(an=a[n],bn=b[n],a0=a[0]);
```

$$an = 0,\ bn = -\frac{6}{n\sim\pi},\ a0 = 6 \tag{4.9}$$

```
> serie_trig:=a[0]/2+ Sum(a[n]*cos(w[n]*t)+b[n]*sin(w[n]*t),n=1..
  infinity);
```

$$serie\_trig := 3 + \sum_{n\sim=1}^{\infty}\left(-\frac{6\sin\left(\frac{n\sim\pi t}{4}\right)}{n\sim\pi}\right) \tag{4.10}$$

```
> serie_real:=expand(evalc(Re(serie_comp10)));
```

$$serie\_real := 3 - \frac{\sin\left(\frac{3\pi t}{2}\right)}{\pi} - \frac{6\sin\left(\frac{\pi t}{4}\right)}{\pi} - \frac{6}{7}\frac{\sin\left(\frac{7\pi t}{4}\right)}{\pi} - \frac{3}{5}\frac{\sin\left(\frac{5\pi t}{2}\right)}{\pi} - \frac{3\sin\left(\frac{\pi t}{2}\right)}{\pi} - \frac{6}{5}\frac{\sin\left(\frac{5\pi t}{4}\right)}{\pi} - \frac{2\sin\left(\frac{3\pi t}{4}\right)}{\pi} - \frac{3}{2}\frac{\sin(\pi t)\cos(\pi t)}{\pi} - \frac{3}{2}\frac{\sin(\pi t)}{\pi} - \frac{2}{3}\frac{\sin\left(\frac{9\pi t}{4}\right)}{\pi} \tag{4.11}$$

Será que a resposta do Maple está correta? Como pode aparecer um termo com o produto seno e co-seno? Os coeficientes an=0 e aparece um co-seno!!!
Sim, a resposta está correta, porque este termo quando simplificado, resulta:

```
> termo:=convert(3/2*1/Pi*sin(Pi*t)*cos(Pi*t),sin);
```

$$termo := \frac{3}{4}\,\frac{\sin(2\,\pi\,t)}{\pi} \tag{4.12}$$

Somente para conferir os cálculos efetuados acima:

```
> serie_trig:=a[0]/2+ Sum(a[n]*cos(w[n]*t)+b[n]*sin(w[n]*t),n=0..
  infinity);
```

$$serie\_trig := 3 + \sum_{n\sim=0}^{\infty}\left(-\frac{6\sin\left(\frac{n\sim\pi\,t}{4}\right)}{n\sim\pi}\right) \tag{4.13}$$

Aproximar a função f(x) com a série truncada n=0..10 termos.

```
> f_ap(t):=a[0]/2+ sum(a[n]*cos(w[n]*t)+b[n]*sin(w[n]*t),n=1..10);
```

$$f\_ap(t) := 3 - \frac{\sin\left(\frac{3\,\pi\,t}{2}\right)}{\pi} - \frac{6\sin\left(\frac{\pi\,t}{4}\right)}{\pi} - \frac{6}{7}\,\frac{\sin\left(\frac{7\,\pi\,t}{4}\right)}{\pi} - \frac{3}{5}\,\frac{\sin\left(\frac{5\,\pi\,t}{2}\right)}{\pi} - \frac{3\sin\left(\frac{\pi\,t}{2}\right)}{\pi} - \frac{6}{5}\,\frac{\sin\left(\frac{5\,\pi\,t}{4}\right)}{\pi} - \frac{2\sin\left(\frac{3\,\pi\,t}{4}\right)}{\pi} - \frac{3}{4}\,\frac{\sin(2\,\pi\,t)}{\pi} - \frac{3}{2}\,\frac{\sin(\pi\,t)}{\pi} - \frac{2}{3}\,\frac{\sin\left(\frac{9\,\pi\,t}{4}\right)}{\pi} \tag{4.14}$$

Espectro de freqüência (espectro de amplitude) - exemplo 2:

é o conjunto de pontos **(n**w_0**,** $|C_n|$ **) ou (**w_n**,** $|C_n|$ **), onde n é um número inteiro.**

Cálculo das amplitudes, para o espectro de freqüência

```
> C[n];moduloCn:=abs(C[n]);C[0]:=3;
```

$$\frac{3\,I}{n\sim\pi} \tag{5.1}$$

$$moduloCn := \frac{3}{\pi\,|n\sim|}$$

$$C_0 := 3$$

```
> for i from -10 to -1 do C[i]:=eval(subs(n=i,moduloCn)) :  w[i]:=
  subs(n=i,w[n]):od:             for i from 1 to 10 do C[i]:=eval
```

```
  (subs(n=i,moduloCn)):  w[i]:=subs(n=i,w[n]):od:
> for k from -10 to -1 do lprint([k],freq=w[k],amplitude=C[k])
  :od:for k from 1 to 10 do lprint([k],freq=w[k],amplitude=C[k])
  :od;
[-10], freq = -5/2*Pi, amplitude = 3/10/Pi
[-9], freq = -9/4*Pi, amplitude = 1/3/Pi
[-8], freq = -2*Pi, amplitude = 3/8/Pi
[-7], freq = -7/4*Pi, amplitude = 3/7/Pi
[-6], freq = -3/2*Pi, amplitude = 1/2/Pi
[-5], freq = -5/4*Pi, amplitude = 3/5/Pi
[-4], freq = -Pi, amplitude = 3/4/Pi
[-3], freq = -3/4*Pi, amplitude = 1/Pi
[-2], freq = -1/2*Pi, amplitude = 3/2/Pi
[-1], freq = -1/4*Pi, amplitude = 3/Pi
[1], freq = 1/4*Pi, amplitude = 3/Pi
[2], freq = 1/2*Pi, amplitude = 3/2/Pi
[3], freq = 3/4*Pi, amplitude = 1/Pi
[4], freq = Pi, amplitude = 3/4/Pi
[5], freq = 5/4*Pi, amplitude = 3/5/Pi
[6], freq = 3/2*Pi, amplitude = 1/2/Pi
[7], freq = 7/4*Pi, amplitude = 3/7/Pi
[8], freq = 2*Pi, amplitude = 3/8/Pi
[9], freq = 9/4*Pi, amplitude = 1/3/Pi
[10], freq = 5/2*Pi, amplitude = 3/10/Pi
> g[0]:=line([0,0],[0,3],color=red, thickness=3):
> for i from -10 to -1 do g[i]:=line([w[i],0],[w[i],C[i]], color=
  red, thickness=3):od:          for i from 1 to 10 do g[i]:=line(
  [w[i],0],[w[i],C[i]], color=red, thickness=3):od:
> Graf2:=plots[display]({g[0],g[-1],g[-2],g[-3],g[-4],g[-5],g[-6],g
  [-7],g[-8],g[-9],g[-10],g[1],g[2],g[3],g[4],g[5],g[6],g[7],g[8],g
  [9],g[10]},titlefont=[COURIER,DEFAULT,12],labels=["w[n]","|C[n]
  |"],labelfont=[COURIER,DEFAULT,10],axesfont=[COURIER,DEFAULT,10],
  title=`Espectro de Amplitude,n=-10..10`):Graf2;
```



Atenção: geralmente a aplitude =|C0| é considerada dividida por dois. Então ampo=1.5 e não 3, como aparece no espectro acima.

==
======FIM===================